NAG Library Routine Document

F07HGF (DPBCON)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07HGF (DPBCON) estimates the condition number of a real symmetric positive definite band matrix A, where A has been factorized by F07HDF (DPBTRF).

2 Specification

```
SUBROUTINE F07HGF (UPLO, N, KD, AB, LDAB, ANORM, RCOND, WORK, IWORK,       &
                   INFO)

INTEGER             N, KD, LDAB, IWORK(N), INFO
REAL (KIND=nag_wp) AB(LDAB,*), ANORM, RCOND, WORK(3*N)
CHARACTER(1)        UPLO
```

The routine may be called by its LAPACK name ***dpbcon***.

3 Description

F07HGF (DPBCON) estimates the condition number (in the 1-norm) of a real symmetric positive definite band matrix A:

$$\kappa_1(A) = \|A\|_1 \left\|A^{-1}\right\|_1.$$

Since A is symmetric, $\kappa_1(A) = \kappa_\infty(A) = \|A\|_\infty \left\|A^{-1}\right\|_\infty$.

Because $\kappa_1(A)$ is infinite if A is singular, the routine actually returns an estimate of the **reciprocal** of $\kappa_1(A)$.

The routine should be preceded by a call to F06REF to compute $\|A\|_1$ and a call to F07HDF (DPBTRF) to compute the Cholesky factorization of A. The routine then uses Higham's implementation of Hager's method (see Higham (1988)) to estimate $\left\|A^{-1}\right\|_1$.

4 References

Higham N J (1988) FORTRAN codes for estimating the one-norm of a real or complex matrix, with applications to condition estimation *ACM Trans. Math. Software* **14** 381–396

5 Arguments

1: UPLO – CHARACTER(1) *Input*

On entry: specifies how A has been factorized.

UPLO = 'U'
$A = U^{\mathrm{T}}U$, where U is upper triangular.

UPLO = 'L'
$A = LL^{\mathrm{T}}$, where L is lower triangular.

Constraint: UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

3: KD – INTEGER *Input*

On entry: k_d, the number of superdiagonals or subdiagonals of the matrix A.

Constraint: KD ≥ 0.

4: AB(LDAB, $*$) – REAL (KIND=nag_wp) array *Input*

Note: the second dimension of the array AB must be at least max(1, N).

On entry: the Cholesky factor of A, as returned by F07HDF (DPBTRF).

5: LDAB – INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F07HGF (DPBCON) is called.

Constraint: LDAB $\geq$ KD + 1.

6: ANORM – REAL (KIND=nag_wp) *Input*

On entry: the 1-norm of the **original** matrix A, which may be computed by calling F06REF with its argument NORM = '1'. ANORM must be computed either **before** calling F07HDF (DPBTRF) or else from a **copy** of the original matrix A.

Constraint: ANORM ≥ 0.0.

7: RCOND – REAL (KIND=nag_wp) *Output*

On exit: an estimate of the reciprocal of the condition number of A. RCOND is set to zero if exact singularity is detected or the estimate underflows. If RCOND is less than ***machine precision***, A is singular to working precision.

8: WORK(3 $\times$ N) – REAL (KIND=nag_wp) array *Workspace*

9: IWORK(N) – INTEGER array *Workspace*

10: INFO – INTEGER *Output*

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO $= -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

7 Accuracy

The computed estimate RCOND is never less than the true value ρ, and in practice is nearly always less than 10ρ, although examples can be constructed where RCOND is much larger.

8 Parallelism and Performance

F07HGF (DPBCON) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

A call to F07HGF (DPBCON) involves solving a number of systems of linear equations of the form $Ax = b$; the number is usually 4 or 5 and never more than 11. Each solution involves approximately $4nk$ floating-point operations (assuming $n \gg k$) but takes considerably longer than a call to F07HEF (DPBTRS) with one right-hand side, because extra care is taken to avoid overflow when A is approximately singular.

The complex analogue of this routine is F07HUF (ZPBCON).

10 Example

This example estimates the condition number in the 1-norm (or ∞-norm) of the matrix A, where

$$A = \begin{pmatrix} 5.49 & 2.68 & 0.00 & 0.00 \\ 2.68 & 5.63 & -2.39 & 0.00 \\ 0.00 & -2.39 & 2.60 & -2.22 \\ 0.00 & 0.00 & -2.22 & 5.17 \end{pmatrix}.$$

Here A is symmetric and positive definite, and is treated as a band matrix, which must first be factorized by F07HDF (DPBTRF). The true condition number in the 1-norm is 74.15.

10.1 Program Text

```
    Program f07hgfe

!     F07HGF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: dlansb => f06ref, dpbcon, dpbtrf, nag_wp, x02ajf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: anorm, rcond
      Integer                          :: i, info, j, kd, ldab, n
      Character (1)                    :: uplo
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: ab(:,:), work(:)
      Integer, Allocatable             :: iwork(:)
!     .. Intrinsic Procedures ..
      Intrinsic                        :: max, min
!     .. Executable Statements ..
      Write (nout,*) 'F07HGF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, kd
      ldab = kd + 1
      Allocate (ab(ldab,n),work(3*n),iwork(n))

!     Read A from data file

      Read (nin,*) uplo
```

```
      If (uplo=='U') Then
        Do i = 1, n
          Read (nin,*)(ab(kd+1+i-j,j),j=i,min(n,i+kd))
        End Do
      Else If (uplo=='L') Then
        Do i = 1, n
          Read (nin,*)(ab(1+i-j,j),j=max(1,i-kd),i)
        End Do
      End If

!     Compute norm of A

!     f06ref is the NAG name equivalent of the LAPACK auxiliary dlansb
      anorm = dlansb('1-norm',uplo,n,kd,ab,ldab,work)

!     Factorize A
!     The NAG name equivalent of dpbtrf is f07hdf
      Call dpbtrf(uplo,n,kd,ab,ldab,info)

      Write (nout,*)
      If (info==0) Then

!       Estimate condition number
!       The NAG name equivalent of dpbcon is f07hgf
        Call dpbcon(uplo,n,kd,ab,ldab,anorm,rcond,work,iwork,info)

        If (rcond>=x02ajf()) Then
          Write (nout,99999) 'Estimate of condition number =',                &
            1.0_nag_wp/rcond
        Else
          Write (nout,*) 'A is singular to working precision'
        End If
      Else
        Write (nout,*) 'A is not positive definite'
      End If

99999 Format (1X,A,1P,E10.2)
    End Program f07hgfe
```

10.2 Program Data

```
F07HGF Example Program Data
  4  1                        :Values of N and KD
  'L'                         :Value of UPLO
  5.49
  2.68   5.63
        -2.39   2.60
               -2.22   5.17   :End of matrix A
```

10.3 Program Results

```
 F07HGF Example Program Results

 Estimate of condition number =  7.42E+01
```